
Appendix B. SAS syntax to create MSM variables from the sexual behavior component of the 
National Health and Nutrition Examination Survey, 1999-2014 

*Download XPT files, save into a temporary folder denoted as [temp folder 
path]; 
*In the libname statement, set the libref to a permanent folder to  
store the (XPT files converted to) SAS files denoted as [permanent folder 
path]; 
libname msm '[permanent folder path]'; 
*Convert XPT to SAS file for the eight NHANES cycles; 
*NHANES cycle 1999-2000; 
libname temp xport '[temp folder path]\SXQ.xpt'; 
proc copy inlib=temp outlib=msm; 
run; 
*NHANES cycle 2001-2002; 
libname temp xport '[temp folder path]\SXQ_B.xpt'; 
proc copy inlib=temp outlib=msm; 
run; 
*NHANES cycle 2003-2004; 
libname temp xport '[temp folder path]\SXQ_C.xpt'; 
proc copy inlib=temp outlib=msm; 
run; 
*NHANES cycle 2005-2006; 
libname temp xport '[temp folder path]\SXQ_D.xpt'; 
proc copy inlib=temp outlib=msm; 
run; 
*NHANES cycle 2007-2008; 
libname temp xport '[temp folder path]\SXQ_E.xpt'; 
proc copy inlib=temp outlib=msm; 
run; 
*NHANES cycle 2009-2010; 
libname temp xport '[temp folder path]\SXQ_F.xpt'; 
proc copy inlib=temp outlib=msm; 
run; 
*NHANES cycle 2011-2012; 
libname temp xport '[temp folder path]\SXQ_G.xpt'; 
proc copy inlib=temp outlib=msm; 
run; 
*NHANES cycle 2013-2014; 
libname temp xport '[temp folder path]\SXQ_H.xpt'; 
proc copy inlib=temp outlib=msm; 
run; 
 
*NHANES cycle 1999-2000; 
data yr0; 
 set msm.SXQ; 
 yr = 0; 
 SQ1 = SXQ020; *ever had sex; 
 SQ2 = .;    *ever had sex with a man; 
 SQ3 = SXQ200; *lifetime male partners; 
 SQ4 = SXQ220; *past 12 months male partners; 
run; 
*NHANES cycle 2001-2002; 
data yr1; 
 set msm.SXQ_B; 
 yr = 1; 



 SQ1 = SXQ020; *ever had sex; 
 SQ2 = .;    *ever had sex with a man; 
 SQ3 = SXQ200; *lifetime male partners; 
 SQ4 = SXQ220; *past 12 months male partners; 
run; 
*NHANES cycle 2003-2004; 
data yr3; 
 set msm.SXQ_C; 
 yr = 3; 
 SQ1 = SXQ020; *ever had sex; 
 SQ2 = .;    *ever had sex with a man; 
 SQ3 = SXQ200; *lifetime male partners; 
 SQ4 = SXQ220; *past 12 months male partners; 
run; 
*NHANES cycle 2005-2006; 
data yr5; 
 set msm.SXQ_D; 
 yr = 5; 
 SQ1 = SXQ021; *ever had sex; 
 SQ2 = .;    *ever had sex with a man; 
 SQ3 = SXQ410; *lifetime male partners; 
 SQ4 = SXQ550; *past 12 months male partners; 
run; 
*NHANES cycle 2007-2008; 
data yr7; 
 set msm.SXQ_E; 
 yr = 7; 
 SQ1 = SXQ021; *ever had sex; 
 SQ2 = .;    *ever had sex with a man; 
 SQ3 = SXQ410; *lifetime male partners; 
 SQ4 = SXQ550; *past 12 months male partners; 
run; 
*NHANES cycle 2009-2010; 
data yr9; 
 set msm.SXQ_F; 
 yr = 9; 
 SQ1 = SXD021; *ever had sex; 
 SQ2 = SXQ809; *ever had sex with a man; 
 SQ3 = SXQ410; *lifetime male partners; 
      SQ4 = SXQ550; *past 12 months male partners; 
run; 
*NHANES cycle 2011-2012; 
data yr11; 
 set msm.SXQ_G; 
 yr = 11; 
 SQ1 = SXD021; *ever had sex; 
 SQ2 = SXQ809; *ever had sex with a man; 
 SQ3 = SXQ410; *lifetime male partners; 
 SQ4 = SXQ550; *past 12 months male partners; 
run; 
*NHANES cycle 2013-2014; 
data yr13; 
 set msm.SXQ_H; 
 yr = 13; 
 SQ1 = SXD021; *ever had sex; 
 SQ2 = SXQ809; *ever had sex with a man; 
 SQ3 = SXQ410; *lifetime male partners; 



 SQ4 = SXQ550; *past 12 months male partners; 
run; 
data SXQdata; 
 set yr0 yr1 yr3 yr5 yr7 yr9 yr11 yr13; 
 *MSM-Ever; 
 if yr in (0, 1, 3, 5, 7) then do; 
  if SQ3 > 0 and SQ3 < 77777 then msmever = 1; 
  else if SQ3 = 0 then msmever = 0; 
  else if SQ3 = . or SQ3 in (77777, 99999) then msmever = .; 
 end;  
 else if yr in (9, 11, 13) then do; 
  if SQ1 = 1 and SQ2 = 2 then msmever = 0; 
  else if SQ1 = 1 and SQ2 = 1 then msmever = 1;  
  else if SQ3 = 0 then msmever = 0; 
  else if SQ3 = . or SQ3 in (77777, 99999) then msmever = .; 
 end; 
 *MSM-Current;  
 if msmever = 0 then msmcurrent = 0; 
 else if msmever = 1 and (SQ4 > 0 and SQ4 < 77777) then msmcurrent = 1; 
 else if SQ4 = . or SQ4 in (77777, 99999) then msmcurrent = .; 
 *MSM-Past; 
 if msmever = 0 then msmpast = 0; 
 else if msmever = 1 and SQ4 = 0 then msmpast = 1; 
 else if SQ4 = . or SQ4 in (77777, 99999) then msmpast = .; 
run; 
	
  


